
1. Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iwd-TTCN3-B2013-03_D15wk10’.

3. Corrections required for POS ATS
Change 1 – Missing IMS default handling
	Function name
	f_MTC_ConnectPTCs_POS ()

	Reason for change
	2.   Missing imports from IMS components in the MTC_Main_POS module.
3.   IMS PTC is not connected.

	Summary of change
	1.   Added missing imports.
2.   Connected the IMS PTC.

	TTCN module
	POS\MTC\MTC_Main_POS.ttcn

	MCC160 Comment
	


Before change

/******************************************************************************/
// @copyright   Copyright Notification
//   No part may be reproduced except as authorized by written permission.
//   The copyright and the foregoing restriction extend to reproduction in all media.
//   Trademark 2015, 3GPP Organizational Partners (ARIB, ATIS, CCSA, ETSI, TSDSI, TTA, TTC).
//   All rights reserved.
// @version: IWD_15wk10
// $Date: 2013-07-02 01:35:04 +0200 (Tue, 02 Jul 2013) $
// $Rev: 9069 $
/******************************************************************************/
module MTC_Main_POS {
  import from MTC_Base all;
  import from MTC_Component_POS all;
  import from MTC_UpperTester all;
  import from UpperTesterSystemCmds all;
  // import from EUTRA_Component all;
  import from EUTRA_POS_Component all;
  import from UTRAN_Component all;
  import from UTRAN_POS_Component all;
  import from NasEmu_Component all;
  import from NasEmu all;      // f_NASEMU_MainLoop
  import from IP_PTC_Component all;
  import from IP_PTC_Main all;    // f_IP_PTC_MainLoop
  import from PosSystem_ASP_TypeDefs all;
  import from TestcaseProperties all;
  //=============================================================================
  // TEMPLATES
  //-----------------------------------------------------------------------------
...

  //=============================================================================
  // POS MTC FUNCTIONS
  //-----------------------------------------------------------------------------
  //============================================================================
  /*
   * @desc      Connect all PTCs (positioning test cases)
   * @param     p_Eutra EUTRA_PTC
   * @param     p_Utran UTRAN_PTC
   * @status    APPROVED (POS)
   */
  function f_MTC_ConnectPTCs_POS(EUTRA_POS_PTC p_Eutra,
                                 UTRAN_POS_PTC p_Utran) runs on MTC_POS
  {
    var IP_PTC v_IP_PTC;
    var NASEMU_PTC v_NASEMU_PTC;
    var boolean v_MapPOSSYS := f_GetTestcaseAttrib_SpecificMTC_SatPort(testcasename());
    vc_Components.Eutra := p_Eutra;
    vc_Components.Utran := p_Utran;
    // ***** MTC *****
    f_MTC_BASE_Map();
    if (v_MapPOSSYS) {
      map(mtc:POSSYS, system:POSSYS);
    }
    // ***** EUTRA PTC *****
    if (p_Eutra != null) {
      f_EUTRA_POS_PTC_Map(p_Eutra);                             // map EUTRA system ports
      v_NASEMU_PTC := f_NasEmu_CreateConnectAndMap(p_Eutra);    // create NasEmu, map system ports and connect with EUTRA PTC
      v_NASEMU_PTC.start(f_NASEMU_MainLoop());                  // start NASEMU
      v_IP_PTC := f_IP_PTC_CreateAndMap();                      // Assumption: IP handling only needed for EUTRA
      connect(p_Eutra:IP, v_IP_PTC:EUTRA_CTRL);
      v_IP_PTC.start(f_IP_PTC_MainLoop());
      connect(mtc:PTC_Ut    [tsc_MTC_PortIndex_EUTRA], p_Eutra:UT);
      connect(mtc:PTC_POSSYS[tsc_MTC_PortIndex_EUTRA], p_Eutra:POSSYS);
    }
    // ***** UTRAN PTC *****
    if (p_Utran != null) {
      f_UTRAN_PTC_Map(p_Utran);
      connect(mtc:PTC_Ut    [tsc_MTC_PortIndex_UTRAN], p_Utran:UT);
      connect(mtc:PTC_POSSYS[tsc_MTC_PortIndex_UTRAN], p_Utran:POSSYS);
    }
    f_UT_InitialPowerOffUE(Ut); /* @sic R5s120520 sic@ */
  }
...
After change
/******************************************************************************/
// @copyright   Copyright Notification
//   No part may be reproduced except as authorized by written permission.
//   The copyright and the foregoing restriction extend to reproduction in all media.
//   Trademark 2015, 3GPP Organizational Partners (ARIB, ATIS, CCSA, ETSI, TSDSI, TTA, TTC).
//   All rights reserved.
// @version: IWD_15wk10
// $Date: 2013-07-02 01:35:04 +0200 (Tue, 02 Jul 2013) $
// $Rev: 9069 $
/******************************************************************************/
module MTC_Main_POS {
  import from MTC_Base all;
  import from MTC_Component_POS all;
  import from MTC_UpperTester all;
  import from UpperTesterSystemCmds all;
  // import from EUTRA_Component all;
  import from EUTRA_POS_Component all;
  import from UTRAN_Component all;
  import from UTRAN_POS_Component all;
  import from NasEmu_Component all;
  import from NasEmu all;      // f_NASEMU_MainLoop
  import from IP_PTC_Component all;
  import from IP_PTC_Main all;    // f_IP_PTC_MainLoop
  import from IMS_Component all;
  import from IMS_DefaultHandler all;       // f_IMS_DefaultHandler
  import from PosSystem_ASP_TypeDefs all;
  import from TestcaseProperties all;
  import from CommonDefs all;
  //=============================================================================
  // TEMPLATES
  //-----------------------------------------------------------------------------
... 

   //=============================================================================
  // POS MTC FUNCTIONS
  //-----------------------------------------------------------------------------
  //============================================================================
  /*
   * @desc      Connect all PTCs (positioning test cases)
   * @param     p_Eutra EUTRA_PTC
   * @param     p_Utran UTRAN_PTC
   * @status    APPROVED (POS)
   */
  function f_MTC_ConnectPTCs_POS(EUTRA_POS_PTC p_Eutra,
                                 UTRAN_POS_PTC p_Utran,
                                 IMS_PTC p_ImsPdn1,
                                 IMS_PTC p_ImsPdn2) runs on MTC_POS
  {
    var IP_PTC v_IP_PTC;
    var IMS_PTC v_ImsPdn1;
    var NASEMU_PTC v_NASEMU_PTC;
    var boolean v_MapPOSSYS := f_GetTestcaseAttrib_SpecificMTC_SatPort(testcasename());
    vc_Components.Eutra := p_Eutra;
    vc_Components.Utran := p_Utran;
    // ***** MTC *****
    f_MTC_BASE_Map();
    if (v_MapPOSSYS) {
      map(mtc:POSSYS, system:POSSYS);
    }
    // ***** EUTRA PTC *****
    if (p_Eutra != null) {
      f_EUTRA_POS_PTC_Map(p_Eutra);                             // map EUTRA system ports
      v_NASEMU_PTC := f_NasEmu_CreateConnectAndMap(p_Eutra);    // create NasEmu, map system ports and connect with EUTRA PTC
      v_NASEMU_PTC.start(f_NASEMU_MainLoop());                  // start NASEMU
      v_IP_PTC := f_IP_PTC_CreateAndMap();                      // Assumption: IP handling only needed for EUTRA
      connect(p_Eutra:IP, v_IP_PTC:EUTRA_CTRL);
      v_IP_PTC.start(f_IP_PTC_MainLoop());
      connect(mtc:PTC_Ut    [tsc_MTC_PortIndex_EUTRA], p_Eutra:UT);
      connect(mtc:PTC_POSSYS[tsc_MTC_PortIndex_EUTRA], p_Eutra:POSSYS);
    }
    // ***** Create, connect and start IMS PTC *****
    // IMS for PDN1: if PTC is not created yet the default IMS PTC will be used
    if (p_ImsPdn1 == null) { v_ImsPdn1 := f_IMS_PTC_Create(); }         // => default IMS handling for PDN1
    else                   { v_ImsPdn1 := p_ImsPdn1; }
    connect(v_ImsPdn1:IMS_CTRL,      v_IP_PTC:IMS_CTRL[tsc_Index_PDN1]);
    connect(v_ImsPdn1:IMS_Server,    v_IP_PTC:IMS_Server[tsc_Index_PDN1]);
    connect(v_ImsPdn1:IMS_Client,    v_IP_PTC:IMS_Client[tsc_Index_PDN1]);
    connect(v_ImsPdn1:IPCAN,         p_Eutra:IMS[tsc_Index_PDN1]);
    connect(v_ImsPdn1:MMI,           mtc:PTC_Ut[tsc_MTC_PortIndex_IMS1]);   /* @sic R5-134070: generic handling of UT commands for IMS PTC as well as for the other PTCs sic@ */
    if (p_ImsPdn1 == null) {         /* => we have no test case specific IMS implementation */
      v_ImsPdn1.start(f_IMS_DefaultHandler());
    }
    // IMS for PDN (emergency call)
    if (p_ImsPdn2 != null) {                             // test case specific IMS handling for PDN2
      connect(p_ImsPdn2:IMS_CTRL,      v_IP_PTC:IMS_CTRL[tsc_Index_PDN2]);
      connect(p_ImsPdn2:IMS_Server,    v_IP_PTC:IMS_Server[tsc_Index_PDN2]);
      connect(p_ImsPdn2:IMS_Client,    v_IP_PTC:IMS_Client[tsc_Index_PDN2]);
      connect(p_ImsPdn2:IPCAN,         p_Eutra:IMS[tsc_Index_PDN2]);
      connect(p_ImsPdn2:MMI,           mtc:PTC_Ut[tsc_MTC_PortIndex_IMS2]);   /* @sic R5-134070: generic handling of UT commands for IMS PTC as well as for the other PTCs sic@
                                                                                 @sic R5s130987 sic@ */
    }
    // ***** UTRAN PTC *****
    if (p_Utran != null) {
      f_UTRAN_PTC_Map(p_Utran);
      connect(mtc:PTC_Ut    [tsc_MTC_PortIndex_UTRAN], p_Utran:UT);
      connect(mtc:PTC_POSSYS[tsc_MTC_PortIndex_UTRAN], p_Utran:POSSYS);
      connect(v_ImsPdn1:OtherIPCAN,  p_Utran:IMS[tsc_Index_PDN1]);
    }
    f_UT_InitialPowerOffUE(Ut); /* @sic R5s120520 sic@ */
  }

... 
Change 2 – Correction to all TCs in the POS Testsuite
	TC name
	testcase TC_6_2_1_1_1s () 
Used as example for all TCs to be amended in the POS Testsuite !

	Reason for change
	IMS PTC not considered in any of the POS test cases when connecting PTCs.

	Summary of change
	Defined IMS PTCs and considered them when connecting PTC.

	TTCN module
	POS_Testsuite.ttcn

	MCC160 Comment
	


Before change

/******************************************************************************/
// @copyright   Copyright Notification
//   No part may be reproduced except as authorized by written permission.
//   The copyright and the foregoing restriction extend to reproduction in all media.
//   Trademark 2015, 3GPP Organizational Partners (ARIB, ATIS, CCSA, ETSI, TSDSI, TTA, TTC).
//   All rights reserved.
// @version: IWD_15wk10
/******************************************************************************/
module POS_Testsuite {
  import from EUTRA_POS_Component all;
  import from UTRAN_POS_Component all;
  import from MTC_Component_POS all;
  import from MTC_Main_POS all;
  import from POS_37571_SelectionExpressions all;
  import from Parameters all;
  import from EUTRA_Positioning_Testcases all;
  import from EUTRA_Positioning_Testcases_UTRAN all;
  import from UTRAN_AGNSS_Testcases all;
  //----------------------------------------------------------------------------
  testcase TC_6_2_1_1_1s() runs on MTC_POS system SYSTEM_POS {
    // @purpose
    //   NI-LR Emergency Call: UE-Based A-GNSS Sub-test 1
    var EUTRA_POS_PTC v_EUTRA_POS  := null;
    var UTRAN_POS_PTC v_UTRAN_POS  := null;
    timer t_GuardTimer := int2float(360);
    v_UTRAN_POS := UTRAN_POS_PTC.create alive;
    f_MTC_ConnectPTCs_POS(v_EUTRA_POS, v_UTRAN_POS);
    v_UTRAN_POS.start(f_TC_6_2_1_1_1s_UTRAN());
    t_GuardTimer.start;
    f_MTC_MainLoop(t_GuardTimer);
  }
  //----------------------------------------------------------------------------
...
... 

...

After change

/******************************************************************************/
// @copyright   Copyright Notification
//   No part may be reproduced except as authorized by written permission.
//   The copyright and the foregoing restriction extend to reproduction in all media.
//   Trademark 2015, 3GPP Organizational Partners (ARIB, ATIS, CCSA, ETSI, TSDSI, TTA, TTC).
//   All rights reserved.
// @version: IWD_15wk10
/******************************************************************************/
module POS_Testsuite {
  import from EUTRA_POS_Component all;
  import from UTRAN_POS_Component all;
  import from MTC_Component_POS all;
  import from MTC_Main_POS all;
  import from IMS_Component all;
  import from POS_37571_SelectionExpressions all;
  import from Parameters all;
  import from EUTRA_Positioning_Testcases all;
  import from EUTRA_Positioning_Testcases_UTRAN all;
  import from UTRAN_AGNSS_Testcases all;
  //----------------------------------------------------------------------------
  testcase TC_6_2_1_1_1s() runs on MTC_POS system SYSTEM_POS {
    // @purpose
    //   NI-LR Emergency Call: UE-Based A-GNSS Sub-test 1
    var EUTRA_POS_PTC v_EUTRA_POS  := null;
    var UTRAN_POS_PTC v_UTRAN_POS  := null;
 
 var IMS_PTC       v_IMS1       := null;
    var IMS_PTC       v_IMS2       := null;
    timer t_GuardTimer := int2float(360);
    v_UTRAN_POS := UTRAN_POS_PTC.create alive;
    f_MTC_ConnectPTCs_POS(v_EUTRA_POS, v_UTRAN_POS, v_IMS1, v_IMS2);
    v_UTRAN_POS.start(f_TC_6_2_1_1_1s_UTRAN());
    t_GuardTimer.start;
    f_MTC_MainLoop(t_GuardTimer);
  }
  //----------------------------------------------------------------------------
...
...

...
